
Abstract. In this paper, we show efficient implementations of binary
field multiplication over ARMv8. We exploit an advanced 64-bit poly-
nomial multiplication (PMULL) supported by ARMv8 and conduct multi-
ple levels of asymptotically faster Karatsuba multiplication. Finally, our
method conducts binary field multiplication within 57 clock cycles for
B-251. Our proposed method on ARMv8 improves the performance by
a factor of 5.5 times than previous techniques on ARMv7.

1 Introduction

Since binary field multiplication is an important component of elliptic curve
cryptography and authenticated encryption, many researches have studied the
high speed implementation of binary field multiplication in software engineering.
The typical binary field multiplication over embedded processor may compute
the results with bitwise-xor and logical shift operations. The other more clever
approach exploits the look-up table by calculating the part of results in advance
[6, 7, 10, 8, 9]. Recently, many modern embedded processors adopt the advanced
built-in binary field multiplication. ARMv7 supports VMULL.P8 operation which
can compute eight 8-bit wise polynomial multiplications with single instruc-
tion. In [2], author shows that efficient implementation techniques to construct
the 64-bit binary field multiplication with the VMULL.P8 operation. After then
multiple levels of Karatsuba multiplication is applied to several binary field mul-
tiplications including F2251 , F2283 and F2571 . The most recent processor, ARMv8,

supports PMULL operation which can compute 64-bit wise polynomial multipli-
cation with single instruction. In [3], author shows that compact implementa-
tion of GCM based authenticated encryption with the PMULL operation. Since
the 64-bit multiplication is quite fast enough for 128-bit multiplication, they
avoid Karatsuba multiplication. The detailed 128-bit polynomial multiplication
is available in Algorithm 1. The implementations achieved 11 times faster results
than ARMv7. However, the paper does not show binary field multiplication for
long length operands. The long operands are required to compute ECC based
cryptography. In this paper, we present efficient implementations of long bits bi-
nary field multiplication. We applied multiple levels of Karatsuba multiplication
and achieved 5.5 times faster than ARMv7 implementations.

Algorithm 1 128-bit Polynomial Multiplication (mul128 p64)

Require: 128-bit Operands A, B.
Ensure: 256-bit Result C.
1: ld1.16b {v2}, [x2] {Load 128-bit Operand A}
2: ld1.16b {v3}, [x1] {Load 128-bit Operand B}
3: movi.16b v6, #0 {Clear Reg}
4: pmull v0.1q, v2.1d, v3.1d {CL ← A[63:0] ·B[63:0]}
5: pmull2 v1.1q, v2.2d, v3.2d {CH ← A[127:64] ·B[127:64]}
6: ext.16b v4, v3, v3, #8 {Shuffle Operand B}
7: pmull v5.1q, v2.1d, v4.1d {CM ← A[63:0] ·B[127:64]}
8: pmull2 v4.1q, v2.2d, v4.2d {CM′ ← A[127:64] ·B[63:0]}
9: eor.16b v4, v4, v5 {CM ← CM ⊕ CM′}
10: ext.16b v5, v6, v4, #8 {Align Result}
11: eor.16b v0, v0, v5 {CL[127:64] ← CL[127:64] ⊕ CM [63:0]}
12: ext.16b v5, v4, v6, #8 {Align Result}
13: eor.16b v1, v1, v5 {CH[63:0] ← CH[63:0] ⊕ CM [127:64]}
14: st1.16b {v0,v1}, [x0] {Return 256-bit Result C}

The remainder of this paper is organized as follows. In Section 2, we recap
the target ARM processor and Karatsuba method. In Section 3, we propose the
efficient binary field multiplication. In Section 4, we evaluate the performance
of proposed methods in terms of clock cycles. Finally, Section 5 concludes the
paper.

2 Related Works

2.1 ARM Processor

ARM processor is a well known family of RISC processor architectures intro-
duced in 1985 [11]. The most recent version, ARMv8, supports both 32-bit and
64-bit processing. The 32-bit ARMv8 architecture is known as AArch32, while



the 64-bit is known as AArch64. An ARMv8 processor can support both, al-
lowing the execution of 32-bit and 64-bit applications. ARM processors sup-
port a single-instruction multiple-data (SIMD) module called the NEON engine.
AArch32 features sixteen 32-bit registers (R0-R15) and sixteen 128-bit NEON
registers (Q0-Q15). The NEON registers can also be viewed as pairs of 64-bit
registers (D0-D32). For example, D0 and D1 are the lower and higher parts of Q0,
respectively. AArch64 features thirty two 64-bit registers (X0-X31) and thirty
two 128-bit NEON registers (V0-V31). The NEON registers can no longer be
viewed as pairs of 64-bit registers. From ARMv8, two polynomial dedicated in-
structions, PMULL and PMULL2, are available. Both of which carry out a single
64-bit multiplication. In both cases, the inputs are 128-bit registers. Their differ-
ence is that in PMULL the lower 64-bit parts of the inputs are used as operands,
while in PMULL2 the higher 64-bit parts are used [3].

2.2 Karatsuba Algorithm

The basic idea of Karatsuba multiplication is to split a multiplication of two
s words operands into three multiplications of size s

2 , which is possible at the
expense of some additions [5]. Taking the multiplication of s words operands
A and B as an example, we represent the operands as A = AH · 2

s
2 + AL and

B = BH · 2
s
2 +BL. The multiplication P = A ·B can be computed according to

the Equation 1.

AH ·BH · 2s + [(AH +AL)(BH +BL)−AH ·BH −AL ·BL] · 2
s
2 +AL ·BL

Karatsuba method roughly executes 3s2

4 mul instructions to multiply two s-
word operands [4]. Recently, the refined Karatsuba’s algorithm from a Crypto
2009 paper by Bernstein [1] makes efficient use of the available registers to keep
the low overheads from load and store instructions.

3 Proposed Method

3.1 Polynomial Multiplication

Polynomial multiplication can be implemented in ordinary or Karatsuba method.
The Karatsuba multiplication is an efficient approach when size of operand is
long enough than processor’s word. Since the ARMv8 processor is 64-bit archi-
tecture, several parameters are not favorable with Karatsuba multiplication. In
this section, we investigate the proper size of minimum Karatsuba technique.

Four and Eight Terms of Multiplications The efficient Karatsuba mul-
tiplication techniques are highly relied on the number of terms where term is
(operand size/word size). In this section we firstly explore the two, four and eight
terms of multiplications. For two terms as studied in [3], ordinary multiplication



is efficient than Karatsuba approach. As we can see the comparison results in
Table 1. Karatsuba method can reduce the number of multiplication by 1. How-
ever, more number of other instructions including eor and ext are required.
For this reason, ordinary multiplication is better choice for 128-bit polynomial
multiplication.

Table 1. Comparison of 128-bit polynomial multiplication methods

Instructions pmull eor movi ext

Ordinary 4 3 1 3
Karatsuba 3 6 1 4

In case of long operands (256, 512 and etc), Karatsuba approach is better
choice, because total number of instructions are smaller than ordinary approach.
We use the Karatsuba algorithm for both cases and our solution is a combination
of the Karatsuba algorithm and a multiplier based on PMULL which we have
named the Karatsuba/NEON/PMULL multiplier (KNP).

Four terms (256-bit) polynomial multiplication is executed with 1 level of
Karatsuba multiplication. The inner 128-bit multiplication is readily established
with Algorithm 1. The detailed four terms polynomial multiplication is available
in Algorithm 2. In Step 1 and 2, 256-bit operands are loaded from memory. In
Step 3 and 4, lower parts of operands are bit-wise exclusive-ored with higher
parts. In Step 5, 128-bit register (v14) is cleared. In Step 6, 7 and 8, three 128-
bit wise polynomial multiplications (CL ← AL · BL and CH ← AH · BH and
CM ← (AL ⊕AH) · (BL ⊕BH)) are conducted. The first two input variables of
mul128 p64 are results and third and fourth variables represent operands and the
last three operands indicate temporal storages. From Step 9 to 14, intermediate
results are bit-wise exclusive-ored. At step 15, total 512-bit results are stored
into memory.

Eight terms (512-bit) polynomial multiplication consists of 2 levels of Karat-
suba multiplication. The first level consists of 256-bit polynomial multiplications
and the multiplication is following the Algorithm 2. The detailed 512-bit multi-
plication is written in Algorithm 3. In Step 1 and 2, 256-bit operands are loaded
from memory. In Step 3 and 6, lower parts of operands are bit-wise exclusive-
ored with higher parts. In Step 7, 128-bit register (v14) is cleared. In Step 8,
9 and 10, three 128-bit wise polynomial multiplications (CL ← AL · BL and
CH ← AH · BH and CM ← (AL ⊕ AH) · (BL ⊕ BH)) are conducted. The first
two input variables of mul256 p64 are results and third and fourth variables rep-
resent operands and the last three operands indicate temporal storages. From
Step 11 to 22, intermediate results are bit-wise exclusive-ored. At step 23 and
24, total 1024-bit results are stored into memory.

Three and Six of Multiplications In this section we explore the three, six
and nine terms of multiplications. For three terms, ordinary multiplication is



Algorithm 2 256-bit Polynomial Multiplication (mul256 p64)

Require: 256-bit Operands A, B.
Ensure: 512-bit Result C.
1: ld1.16b {v6, v7}, [x2] {Load 256-bit Operand A}
2: ld1.16b {v9, v10}, [x1] {Load 256-bit Operand B}
3: eor.16b v8, v6, v7 {AL ⊕AH}
4: eor.16b v11, v9, v10 {BL ⊕BH}
5: movi.16b v14, #0 {Clear Reg}
6: mul128 p64 v0, v1, v6, v9, v12, v13, v14 {CL ← AL ·BL}
7: mul128 p64 v2, v3, v7, v10, v12, v13, v14 {CH ← AH ·BH}
8: mul128 p64 v4, v5, v8, v11, v12, v13, v14 {CM ← (AL ⊕AH) · (BL ⊕BH)}
9: eor.16b v4, v4, v0 {CM [127:0] ← CM [127:0] ⊕ CL[127:0]}
10: eor.16b v5, v5, v1 {CM [255:128] ← CM [255:128] ⊕ CL[255:128]}
11: eor.16b v4, v4, v2 {CM [127:0] ← CM [127:0] ⊕ CH[127:0]}
12: eor.16b v5, v5, v3 {CM [255:128] ← CM [255:128] ⊕ CH[255:128]}
13: eor.16b v1, v1, v4 {CL[255:128] ← CL[255:128] ⊕ CM [127:0]}
14: eor.16b v2, v2, v5 {CH[127:0] ← CH[127:0] ⊕ CM [255:128]}
15: st1.16b {v0,v1,v2,v3}, [x0] {Return 512-bit Result C}

the more efficient method than that of Karatsuba. The comparison results are
drawn in Table 2. Three terms of Karatsuba multiplication reduces the number
of multiplication from 9 to 6. However, additional 8 and 5 times of eor and
ext instructions are required. For this reason, ordinary multiplication is better
choice for 192-bit polynomial multiplication.

Table 2. Comparison of 192-bit polynomial multiplication methods

Instructions pmull eor movi ext

Ordinary 6 16 1 8
Karatsuba 9 7 1 3

The detailed three terms of polynomial multiplication is available in Algo-
rithm 4. In Step 1 and 2, 192-bit operands (A and B) are loaded from mem-
ory. By using option 8b, we loaded operands by sequential 64-bit format to
the 128-bit registers, which lefts higher 64-bit as an empty. In Step 3 ∼ 6,
four multiplications including (CL ← A[63:0] · B[63:0], TL ← A[63:0] · B[127:64],
CM ← A[63:0] · B[191:128] and Temp ← A[127:64] · B[63:0]) are conducted. After
then results (A[127:64] · B[63:0]) are added to TL. From Step 8 to 15, remaining
multiplications are conducted and then added to the intermediate results. After
then the results are aligned and accumulated to the intermediate results in Step
17 ∼ 22. Finally, total 384-bit results are stored into memory.

For six terms of polynomial multiplication, we firstly split six terms largely
into two parts consisting of three terms. The two part is computed with 1 level
of Karatsuba multiplication. The inner three-term of multiplication is executed



Algorithm 3 512-bit Polynomial Multiplication

Require: 512-bit Operands A, B.
Ensure: 1024-bit Result C.
1: ld1.16b {v0, v1, v2, v3}, [x2] {Load 512-bit Operand A}
2: ld1.16b {v6, v7, v8, v9}, [x1] {Load 512-bit Operand B}
3: eor.16b v4, v0, v2 {AL[127:0] ⊕AH[127:0]}
4: eor.16b v5, v1, v3 {AL[255:128] ⊕AH[255:128]}
5: eor.16b v10, v6, v8 {BL[127:0] ⊕BH[127:0]}
6: eor.16b v11, v7, v9 {BL[255:128] ⊕BH[255:128]}
7: movi.16b v16, #0 {Clear Reg}
8: mul256 p64 v18, v19, v20, v21, v30, v31, v0, v1, v12, v6, v7, v13,

v14, v15, v16 {CL ← AL ·BL}
9: mul256 p64 v22, v23, v24, v25, v30, v31, v2, v3, v12, v8, v9, v13,

v14, v15, v16 {CH ← AH ·BH}
10: mul256 p64 v26, v27, v28, v29, v30, v31, v4, v5, v12, v10, v11, v13,

v14, v15, v16 {CM ← (AL ⊕AH) · (BL ⊕BH)}
11: eor.16b v26, v26, v18 {CM [127:0] ← CM [127:0] ⊕ CL[127:0]}
12: eor.16b v27, v27, v19 {CM [255:128] ← CM [255:128] ⊕ CL[255:128]}
13: eor.16b v28, v28, v20 {CM [383:256] ← CM [383:256] ⊕ CL[383:256]}
14: eor.16b v29, v29, v21 {CM [511:384] ← CM [511:384] ⊕ CL[511:384]}
15: eor.16b v26, v26, v22 {CM [127:0] ← CM [127:0] ⊕ CH[127:0]}
16: eor.16b v27, v27, v23 {CM [255:128] ← CM [255:128] ⊕ CH[255:128]}
17: eor.16b v28, v28, v24 {CM [383:256] ← CM [383:256] ⊕ CH[383:256]}
18: eor.16b v29, v29, v25 {CM [511:384] ← CM [511:384] ⊕ CH[511:384]}
19: eor.16b v20, v20, v26 {CL[383:256] ← CL[383:256] ⊕ CM [127:0]}
20: eor.16b v21, v21, v27 {CL[511:384] ← CL[511:384] ⊕ CM [255:128]}
21: eor.16b v22, v22, v28 {CH[127:0] ← CH[127:0] ⊕ CM [383:256]}
22: eor.16b v23, v23, v29 {CH[255:128] ← CH[255:128] ⊕ CM [511:384]}
23: st1.16b {v18, v19, v20, v21}, [x0], #64 {Return 512-bit Result C[511:0]}
24: st1.16b {v22, v23, v24, v25}, [x0], #64 {Return 512-bit Result C[1023:512]}



Algorithm 4 192-bit Polynomial Multiplication (mul192 p64)

Require: 192-bit Operands A, B.
Ensure: 384-bit Result C.
1: ld1.8b {v0, v1, v2}, [x2] {Load 192-bit Operand A}
2: ld1.8b {v3, v4, v5}, [x1] {Load 192-bit Operand B}
3: pmull v6.1q, v0.1d, v3.1d {CL ← A[63:0] ·B[63:0]}
4: pmull v9.1q, v0.1d, v4.1d {TL ← A[63:0] ·B[127:64]}
5: pmull v7.1q, v0.1d, v5.1d {CM ← A[63:0] ·B[191:128]}
6: pmull v11.1q, v1.1d, v3.1d {Temp← A[127:64] ·B[63:0]}
7: eor.16b v9, v9, v11 {TL ← TL ⊕ Temp}
8: pmull v11.1q, v1.1d, v4.1d {Temp← A[127:64] ·B[127:64]}
9: eor.16b v7, v7, v11 {CM ← CM ⊕ Temp}
10: pmull v10.1q, v1.1d, v5.1d {TH ← A[127:64] ·B[191:128]}
11: pmull v11.1q, v2.1d, v3.1d {Temp← A[191:128] ·B[63:0]}
12: eor.16b v7, v7, v11 {CM ← CM ⊕ Temp}
13: pmull v11.1q, v2.1d, v4.1d {Temp← A[191:128] ·B[127:64]}
14: eor.16b v10, v10, v11 {TH ← TH ⊕ Temp}
15: pmull v8.1q, v2.1d, v5.1d {CH ← A[191:128] ·B[191:128]}
16: movi.16b v11, #0 {Clear Reg}
17: ext.16b v11, v11, v9 , #8 {Align Result}
18: ext.16b v9, v9, v10 , #8 {Align Result}
19: ext.16b v10, v10, v11, #8 {Align Result}
20: eor.16b v6, v6, v11 {CL[127:64] ← CL[127:64] ⊕ TL[63:0]}
21: eor.16b v7, v7, v9 {CM ← CM ⊕ {TH[63:0]||TL[127:64]}}
22: eor.16b v8, v8, v10 {CH[63:0] ← CH[63:0] ⊕ TH[127:64]}
23: st1.16b {v6,v7,v8}, [x0] {Return 384-bit Result C}



with ordinary multiplication described in Algorithm 4. The detailed six terms
of polynomial multiplication is available in Algorithm 5. In Step 1 ∼ 4, 384-bit
operands are loaded from memory. In Step 5 ∼ 10, lower parts of operands are
bit-wise exclusive-ored with higher parts. In Step 11, 128-bit register (v14) is
cleared. In Step 12, 13 and 14, three 192-bit wise polynomial multiplications
(CL ← AL · BL and CH ← AH · BH and CM ← (AL ⊕ AH) · (BL ⊕ BH))
are conducted. The first six input variables of mul192 p64 are operands and
seventh, eighth and nine-th variables represent temporal storages and the last
three operands indicate results. From Step 15 to 28, intermediate results are
bit-wise exclusive-ored. At step 29 and 30, total 768-bit results are stored into
memory.

3.2 Binary Field Multiplication

s word of binary field multiplication produce values of degree at most 2s − 2,
which must be reduced modulo f(z) = zm + r(z). The usual approach is to
multiply the higher parts by r(z) using shift and xors. For small polynomials
r(z) we can exploit the PMULL instruction to carry out 64-bit multiplication by
r(z).

Curve B-251 The modulo of binary field F2251 is defined by (r(z)= z7 + z4 +
z2 + 1). The detailed reduction method is available in Algorithm 6. In Step 1
∼ 3, intermediate results are shifted by 59 to the right. In Step 4 and 5, 60 ∼
64-th bits are cleared. In Step 6 and 7, intermediate results are shifted by 5 to
the left. In Step 8 ∼ 10, the results are aligned and then the shifted results are
accumulated. In Step 13 and 14, the lower 8-bit of register v15 is set to 0x95. In
Step 15 ∼ 19, the higher intermediate results are multiplied by r(z). In Step 20
∼ 27, the computed results are aligned and then accumulated to intermediate
results. From Step 28 to 41, one more round of reduction process is conducted.

4 Evaluation

In order to test ARMv8 instruction set, we set the development environment
as follows. We used Xcode (ver 6.3.2) as a development IDE and tested the
program over iPad Mini2 (iOS 8.4). The iPad Mini2 supports Apple A7 with
64-bit architecture operated in 1.3GHz. In Table 3, we show performance results
onv various length of polynomial multiplications. For 192-bit, we used ordinary
multiplication and the other parameters exploit the Karatsuba multiplication.

In Table 4, the comparison results of binary field multiplication is available.
Previous works by [2] uses KNV method. This method is quiet slow because it
adopts eight vectorized 8-bit polynomial multiplication namely VMULL. Unlike
previous works, we exploit new 64-bit polynomial multiplication namely PMULL.
This method significantly improves the performance by a factor of 5.5 times than
previous works. However, we couldn’t compare our results with the methods
on same ARMv8 architecture because this is the first implementation for ECC



Algorithm 5 384-bit Polynomial Multiplication

Require: 384-bit Operands A, B.
Ensure: 768-bit Result C.
1: ld1.8b {v0, v1, v2}, [x2], #24 {Load 192-bit Operand A[191:0]}
2: ld1.8b {v3, v4, v5}, [x2], #24 {Load 192-bit Operand A[383:192]}
3: ld1.8b {v9 , v10, v11}, [x1], #24 {Load 192-bit Operand B[191:0]}
4: ld1.8b {v12, v13, v14}, [x1], #24 {Load 192-bit Operand B[383:192]}
5: eor.16b v6, v0, v3 {AL[63:0] ⊕AH[63:0]}
6: eor.16b v7, v1, v4 {AL[127:64] ⊕AH[127:64]}
7: eor.16b v8, v2, v5 {AL[191:128] ⊕AH[191:128]}
8: eor.16b v15, v9, v12 {BL[63:0] ⊕BH[63:0]}
9: eor.16b v16, v10, v13 {BL[127:64] ⊕BH[127:64]}
10: eor.16b v17, v11, v14 {BL[191:128] ⊕BH[191:128]}
11: movi.16b v18, #0 {Clear Reg}
12: mul192 p64 v0, v1, v2, v9, v10, v11, v19, v20, v21, v25, v26, v27

{CL ← AL ·BL}
13: mul192 p64 v3, v4, v5, v12, v13, v14, v19, v20, v21, v28, v29, v30

{CH ← AH ·BH}
14: mul192 p64 normal v6, v7, v8, v15, v16, v17, v19, v20, v21, v0, v1, v2

{CM ← (AL ⊕AH) · (BL ⊕BH)}
15: eor.16b v0, v0, v25 {CM [127:0] ← CM [127:0] ⊕ CL[127:0]}
16: eor.16b v1, v1, v26 {CM [255:128] ← CM [255:128] ⊕ CL[255:128]}
17: eor.16b v2, v2, v27 {CM [383:256] ← CM [383:256] ⊕ CL[383:256]}
18: eor.16b v0, v0, v28 {CM [127:0] ← CM [127:0] ⊕ CH[127:0]}
19: eor.16b v1, v1, v29 {CM [255:128] ← CM [255:128] ⊕ CH[255:128]}
20: eor.16b v2, v2, v30 {CM [383:256] ← CM [383:256] ⊕ CH[383:256]}
21: ext.16b v4, v18, v0, #8 {Align Result}
22: ext.16b v5, v0, v1, #8 {Align Result}
23: ext.16b v6, v1, v2, #8 {Align Result}
24: ext.16b v7, v2, v18, #8 {Align Result}
25: eor.16b v26, v26, v4 {CL[255:191] ← CL[255:191] ⊕ CM [63:0]}
26: eor.16b v27, v27, v5 {CL[383:256] ← CL[383:256] ⊕ CM [191:64]}
27: eor.16b v28, v28, v6 {CH[127:0] ← CH[127:0] ⊕ CM [319:192]}
28: eor.16b v29, v29, v7 {CH[191:128] ← CH[191:128] ⊕ CM [383:320]}
29: st1.16b {v25,v26,v27}, [x0], #48 {Return 384-bit Result C[383:0]}
30: st1.16b {v28,v29,v30}, [x0], #48 {Return 384-bit Result C[767:384]}

Table 3. Performance of polynomial multiplication, (2): two-term

Length 192 256 384 512

Karatsuba - 1-level(2) 1-level(2) 2-level(2)
Clock Cycles 11 21 49 75



Algorithm 6 251-bit Polynomial Multiplication

Require: 502-bit Operands A.
Ensure: 251-bit Result C.
1: ushr.2d v4, v1, #59 {Right Shift by #59}
2: ushr.2d v5, v2, #59 {Right Shift by #59}
3: ushr.2d v6, v3, #59 {Right Shift by #59}
4: shl.2d v1, v1, #5 {Clear High Bits}
5: ushr.2d v1, v1, #5 {Clear High Bits}
6: shl.2d v7, v2, #5 {Left Shift by #5}
7: shl.2d v8, v3, #5 {Left Shift by #5}
8: ext.16b v4, v4, v5, #8 {Align Result}
9: ext.16b v5, v5, v6, #8 {Align Result}
10: ext.16b v6, v6, v14, #8 {Align Result}
11: eor.16b v4, v4, v7 {Accumulate Intermediate Result}
12: eor.16b v5, v5, v8 {Accumulate Intermediate Result}
13: movi.16b v15, #149 {Set 0x95}
14: ushr.2d v15, v15, #56 {Clear High 56-bit}
15: pmull v16.1q, v4.1d, v15.1d {Multiply Modulo}
16: pmull2 v17.1q, v4.2d, v15.2d {Multiply Modulo}
17: pmull v18.1q, v5.1d, v15.1d {Multiply Modulo}
18: pmull2 v19.1q, v5.2d, v15.2d {Multiply Modulo}
19: pmull v20.1q, v6.1d, v15.1d {Multiply Modulo}
20: ext.16b v21, v14, v17, #8 {Align Result}
21: ext.16b v22, v17, v19, #8 {Align Result}
22: ext.16b v23, v19, v14, #8 {Align Result}
23: eor.16b v16, v16, v21 {Accumulate Intermediate Result}
24: eor.16b v18, v18, v22 {Accumulate Intermediate Result}
25: eor.16b v2, v20, v23 {Accumulate Intermediate Result}
26: eor.16b v0, v0, v16 {Accumulate Intermediate Result}
27: eor.16b v1, v1, v18 {Accumulate Intermediate Result}
28: ushr.2d v4, v1, #59 {Right Shift by #59}
29: ushr.2d v5, v2, #59 {Right Shift by #59}
30: shl.2d v1, v1, #5 {Clear High Bits}
31: ushr.2d v1, v1, #5 {Clear High Bits}
32: shl.2d v7, v2, #5 {Left Shift by #5}
33: ext.16b v4, v4, v5, #8 {Align Result}
34: eor.16b v4, v4, v7 {Accumulate Intermediate Result}
35: pmull v16.1q, v4.1d, v15.1d {Multiply Modulo}
36: pmull2 v17.1q, v4.2d, v15.2d {Multiply Modulo}
37: ext.16b v21, v14, v17, #8 {Align Result}
38: ext.16b v22, v17, v14, #8 {Align Result}
39: eor.16b v0, v0, v16 {Accumulate Intermediate Result}
40: eor.16b v0, v0, v21 {Accumulate Intermediate Result}
41: eor.16b v1, v1, v22 {Accumulate Intermediate Result}



friendly binary field multiplication. The recent work by [3] only explores the
short 128-bit binary field multiplication.

Table 4. Comparison results of binary field multiplication

Algorithm Architecture Processor F2251

KNV [2] Cortex-A8 ARMv7 385
KNV [2] Cortex-A9 ARMv7 491
KNV [2] Cortex-A15 ARMv7 317

Proposed Method (KNP) Apple-A7 ARMv8 57

5 Conclusion

In this paper, we show efficient implementation techniques for binary field mul-
tiplication on ARMv8. Our proposed method improves the performance by a
factor of 5.5 times than previous techniques on ARMv7. Our future works are
efficient binary field elliptic curve cryptography with proposed techniques.

3. C. P. Gouvêa and J. López. Implementing gcm on armv8. In Topics in Cryptology—
CT-RSA 2015, pages 167–180. Springer, 2015.

